

ローカル/グローバル最適化を使用した SmartSpice 回路設計

はじめに

SmartSpice のオプティマイザ機能は、回路の変数やパラメータを最適化します。回路設計の分野では、変数または回路パラメータのセットについて、一定の制約の中で（一般的には設計パラメータの下限と上限です）、特定のターゲット目標値を実現することを目指します。初期値とターゲット目標値のセットに基づいて、汎用の最適化エンジンが、目標値に達するまでパラメータ空間上で反復処理を行います。回路は、定常状態、周波数領域、時間領域など、利用可能な任意の解析手法を用いて最適化できます。

最適化には 2 つの種類があります。「性能測定項目の最適化」と「機能の最適化」です。

性能測定項目の最適化では、ある電気的特性が確実に仕様を満たすように、変数およびデバイス / モデル・パラメータを変更します。電気的特性の例として、遅延、立上り / 立下り時間、トリップポイント、最大 / 最小電流、ゲイン、S パラメータのほか、SmartSpice の .MEASURE ステートメントで計算できるその他の回路性能測定項目が挙げられます。

機能の最適化では、回路のある定義済み機能が仕様を満たすように、変数およびパラメータを変更します。例として、計算された曲線を、DC、AC、および過渡解析で得られたターゲット応答に合わせ込む場合が挙げられます。

ローカル最適化法とグローバル最適化法

最適化アルゴリズムは、ローカル最適化法とグローバル最適化法のいずれかに分けられます。ローカル最適化法は、改善のために適切な探索方向を必要とします。概して、ローカル・ミニマムの近傍で堅牢かつ高速です。一方、グローバル最適化法は、ローカル・ミニマムに陥らないようにするため、またグローバル・ミニマムの探索に確率的探索などを使用するため、一般的にローカル法よりも時間がかかります。

すべての最適化法に共通するのは、解の合理的な推定値である開始点です。この開始点、すなわちパラメータの初期値のセットは、過去の経験や、問題に関する知識、または設計空間の範囲を元にした合理的な仮定から導き出すことができます。

ローカル・オプティマイザは、連続するポイントを反復生成し、解が指定された設計条件を満たしたとき、またはそれ以上の改

善が不可能になると終了します。

グローバル・オプティマイザは、すべてのローカル・ミニマムの中で最小の目的関数を探します。ローカル・ミニマムにとらわれることなくグローバル・ミニマムを探索する必要があるため、グローバルな最適化問題は、ローカルな最適化問題よりも複雑です。ローカル・オプティマイザと比較すると、開始点からはるかに離れた設計空間を探索する可能性があります。

SmartSpice では、現在、複数の最適化アルゴリズムを利用できます。ローカル最適化法としては、よく知られている Levenberg-Marquardt 法があり、開始点が解の近くであるときうまく動作します。グローバル最適化法としては、Hooke-Jeeves 法、Simulated Annealing 法、Parallel Tempering 法、遺伝的アルゴリズム、Differential Evolution 法、およびハイブリッド法があります。それぞれの手法の詳細については、『SmartSpice ユーザーズ・マニュアル Volume 2』を参照してください。

設定

グローバル最適化を使用するためのシンタックスは、既存の .MODIF 最適化セクションと同様です。さまざまな最適化ストラテジから 1 つを選択するには、OPTIMIZER オプションを次のように指定します。

```
.OPTION OPTIMIZER=val
```

ここで、val はグローバル・オプティマイザの種類です。次のいずれかを指定します。

- **LM** – Levenberg-Marquardt 法
- **HJ** – Hooke-Jeeves 法
- **SA** – Simulated Annealing 法
- **PT** – Parallel Tempering 法
- **GA** – Genetic Algorithm (遺伝的アルゴリズム)
- **DE** – Differential Evolution 法
- **H** – ハイブリッド・オプティマイザ

デフォルトはLM、つまり、ローカル・オブティマイザのLevenberg-Marquardt法です。

最適化アルゴリズムの設定を行うには、[編集]メニューから[環境設定]ダイアログを開き、[ツール]設定グループの[オブティマイザ]設定ページを使用します。このダイアログでは、停止条件、許容誤差、反復、シミュレーション数、その他のオブティマイザ固有のパラメータなど、最適化に関する各種設定を行えます。たとえば、図1はSimulated Annealingアルゴリズムの最適化設定を示しています。

C₁₃、およびC₂の4つの最適化変数があります。それぞれの開始値は[90nH, 0.5nH, 0.1pF, 25pF]です。ターゲット目標値は、通過帯域で|S₂₁| = -0.5dB、通過帯域の低域側で-55dB (500MHz ~ 600 MHz)、通過帯域の高域側で-60dB (1.4GHz ~ 1.5GHz)です。Levenberg-Marquardt法による最適化を行った後の最終的な素子の値は、[150.5nH, 0.908nH, 0.167pF, 27.7pF]となります。図3は、最適化前、最適化後の|S₂₁|の結果を示しています。

デザイン #1 バンドパス・フィルタ

この例では、通過帯域で0.5dBの等リップル応答を持つ単純なバンドパス・フィルタを設計する場合を考えます。中心周波数は1GHz、バンド幅は10%、インピーダンスは50Ωです。SmartSpice ネットリストを図2に示します。L₁₃、L₂、




図1: 最適化に関する[環境設定]ダイアログ

```

* BAND PASS FILTER with z0=(50,50)
Vin IN 0 ac 1 dc 0
Vout OUT 0 ac 1 dc 0
L1 IN 1 L_13
C1 1 2 C_13
L2 2 0 L_2
C2 2 0 C_2
L3 2 3 L_13
C3 3 OUT C_13

.NET N=2 IN OUT 0 LIN 1001 500Meg 1.5G

.PARAM l_13=154.3nH l_2=0.666nH c_13=0.135pF c_2=39.33pF

.MEASURE NET passband MIN db(sm(out,in)) FROM=950Meg TO=1.05G
.MEASURE NET lower MAX db(sm(out,in)) FROM=500Meg TO=600Meg
.MEASURE NET upper MAX db(sm(out,in)) FROM=1400Meg TO=1500Meg

.MODIF
+ OPTIMIZE L_13=opt(10nH 200nH 90nH)
+ L_2 =opt(0.01nH 10nH 0.5nH)
+ C_13=opt(0.01pF 20pF 0.1pF)
+ C_2 =opt(5pF 300pF 25pF)
+ TARGETS passband=-0.5 lower=-55 upper=-60
+ OPTIONS AVG=1e-8 MEASOFF=1 NUMFUNC=300 NUMITER=300
+ IPLOFF=2 MARQUP=8000

.OPTIONS NOMOD iplot_one
.OPTION OPTIMIZER=LM
.IPLOT db(sm(out,in))
.END
    
```

図2: バンドパス・フィルタのSmartSpiceネットリスト




図3: Levenberg-Marquardt法による最適化前および最適化後の|S21|の結果

デザイン #2 広帯域増幅器のゲイン

この例では、広帯域増幅器のゲインを最適化する場合を考えます。この回路は、すべての抵抗素子と多くのBJTトランジスタ・パラメータを含む120個の設計パラメータを使用します。ここでは、最小ゲインが119、最大ゲインが120、1Hz～10MHzである最終デザインを目標とします。ローカル・オブティマイザのLevenberg-Marquardt法では、満足のいく解を得ることができません。そのため、グローバル・オブティマイザ、ここではSimulated Annealing法を使用します。次のようにオプションを選択します。

```
.OPTION OPTIMIZER=SA
```

SmartSpice ネットリストを図4に示します。Simulated Annealing法による最適化前と最適化後の結果を図5に示します。ゲインは122で平坦であり、10MHzで112に減少します。Simulated Annealing法の代わりに、ハイブリッド最適化ストラテジを選択してSimulated Annealing法、Levenberg-Marquardt法を順に使用した場合、最終結果は低周波で123、10MHzで114となります。ハイブリッド・オブティマイザにおいて、2つ目の最適化ストラテジであるLevenberg-Marquardt法は、1つ目の最適化ストラテジ(この場合はSimulated Annealing法)で見つかった解の近傍でさらにローカル最適化を実行します。

```
EXAMPLE4: RCA3040 WIDEBAND AMPLIFIER

VIN  1 0  DC 0  SIN(0 0.1 50MEG 0.5NS)  AC 1
VEE  3 0  -15
VCC  2 0  15
*
RS2  31 0 1K
RS1  30 1 1K
*
Q1  2  30 5  QNLA
Q2  2  31 6  QNLA
*
Q3  10 5  7  QNLB
Q4  11 6  7  QNLB
*
Q5  14 12 10 QNLC
Q6  15 12 11 QNLC
*
Q7  12 12 13 QNLD
Q8  13 13 0  QNLD
*
Q9  7  8  9  QNL
*
Q10 2  15 16 QNLE
Q11 2  14 17 QNLE
*
R1  5 3  ra
R2  6 3  rb
R3  9 3  rc
R4  8 3  rd
R5  8 0  re
R6  2 14 rf
R7  2 12 rg
R8  2 15 rh
R9  16 0 ri
R10 17 0 rj
*
.MODEL  QNL  NPN(BF=80      RB=100  CCS=2PF  TF=0.3NS  AF=1.0
```

continued...

```

+           KF=5.4E-16 TR=6NS  CJE=3PF  CJC=2PF  VA = 50 )

.MODEL  QNLA  NPN(BF=80      RB=100    CCS=2PF  TF=0.3NS  AF=1.0
+           NF=1.0      BR=2.38  NR=1.0    VJE=1.44  MJE=0.33
+           VJC=0.99    MJC=0.33  VJS=1.39  MJS=0.5   IS=92.5e-16
+           KF=5.4e-16  TR=6NS    CJE=3PF   CJC=2PF   VA=50)

.MODEL  QNLB  NPN(BF=80      RB=100    CCS=2PF  TF=0.3NS  AF=1.0
+           NF=1.0      BR=2.38  NR=1.0    VJE=1.44  MJE=0.33
+           VJC=0.99    MJC=0.33  VJS=1.39  MJS=0.5   IS=92.5e-16
+           KF=5.4e-16  TR=6NS    CJE=3PF   CJC=2PF   VA=50)

.MODEL  QNLC  NPN(BF=80      RB=100    CCS=2PF  TF=0.3NS  AF=1.0
+           NF=1.0      BR=2.38  NR=1.0    VJE=1.44  MJE=0.33
+           VJC=0.99    MJC=0.33  VJS=1.39  MJS=0.5   IS=92.5e-16
+           KF=5.4e-16  TR=6NS    CJE=3PF   CJC=2PF   VA=50)

.MODEL  QNLD  NPN(BF=80      RB=100    CCS=2PF  TF=0.3NS  AF=1.0
+           NF=1.0      BR=2.38  NR=1.0    VJE=1.44  MJE=0.33
+           VJC=0.99    MJC=0.33  VJS=1.39  MJS=0.5   IS=92.5e-16
+           KF=5.4e-16  TR=6NS    CJE=3PF   CJC=2PF   VA=50)

.MODEL  QNLE  NPN(BF=80      RB=100    CCS=2PF  TF=0.3NS  AF=1.0
+           NF=1.0      BR=2.38  NR=1.0    VJE=1.44  MJE=0.33
+           VJC=0.99    MJC=0.33  VJS=1.39  MJS=0.5   IS=92.5e-16
+           KF=5.4e-16  TR=6NS    CJE=3PF   CJC=2PF   VA=50)
*
**** Parameter labels
.PARAM  ra=4.8K  rb=4.8K  rc=811  rd=2.17K  re=820
+       rf=1.32K  rg=4.5K  rh=1.32K  ri=5.25K  rj=5.25K
*
**** MEASURE
.MEASURE AC gain_min MIN vm(16) FROM=1 TO=10e6
.MEASURE AC gain_max MAX vm(16) FROM=1 TO=10e6

.AC  DEC  5 1 10GHZ

.OPTIONS NOMOD iplot_one

**** Optimization specification
.MODIF
+   OPTIMIZE  ra=opt(1K  10K  2K)
+             rb=opt(1K  10K  2K)
+             rc=opt(100 2K  200)
+             rd=opt(500 5K  1K)
+             re=opt(100 2K  500)
+             rf=opt(500 5K  1K)
+             rg=opt(1K  10K  2K)
+             rh=opt(500 5K  12K)
+             ri=opt(1K  10K  2K)
+             rj=opt(1K  10K  2K)

```

```

+           QNL(BF)=OPT( 10 200 80 )
+           QNL(RB)=OPT( 10 500 100 )
+           QNL(CCS)=OPT( 0.5pF 10pF 2pF )
+           QNL(TF)=OPT( 0.1NS 1NS 0.3NS )
+           QNL(AF)=OPT( 0.2 5.0 1.0 )
+           QNL(KF)=OPT( 1e-18 1e-14 5.4e-16 )
+           QNL(TR)=OPT( 2ns 15ns 6ns )
+           QNL(CJE)=OPT( 0.5pF 10pF 3pF )
+           QNL(CJC)=OPT( 0.5pF 10pF 2pF )
+           QNL(VA)=OPT( 10 100 50 )

+           QNLA(BF)=OPT( 10 200 80 )
+           QNLA(RB)=OPT( 10 500 100 )
+           QNLA(CCS)=OPT( 0.5pF 10pF 2pF )
+           QNLA(TF)=OPT( 0.1NS 1NS 0.3NS )
+           QNLA(AF)=OPT( 0.2 5.0 1.0 )
+           QNLA(NF)=OPT( 0.1 3 1.0 )
+           QNLA(BR)=OPT( 0.2 3 2.38 )
+           QNLA(NR)=OPT( 0.5 3 1.0 )
+           QNLA(VJE)=OPT( 0.1 1.5 1.44 )
+           QNLA(MJE)=OPT( 0.1 1.5 0.33 )
+           QNLA(VJC)=OPT( 0.25 2 0.99 )
+           QNLA(MJC)=OPT( 0.1 1 0.33 )
+           QNLA(VJS)=OPT( 0.2 1.5 1.39 )
+           QNLA(MJS)=OPT( 0.1 2 0.5 )
+           QNLA(IS)=OPT( 1e-18 1e-14 92.51e-16 )
+           QNLA(KF)=OPT( 1e-18 1e-14 5.4e-16 )
+           QNLA(TR)=OPT( 2ns 15ns 6NS )
+           QNLA(CJE)=OPT( 0.5pF 10pF 3PF )
+           QNLA(CJC)=OPT( 0.5pF 10pF 2PF )
+           QNLA(VA)=OPT( 10 100 50 )

+           QNLB(BF)=OPT( 10 200 80 )
+           QNLB(RB)=OPT( 10 500 100 )
+           QNLB(CCS)=OPT( 0.5pF 10pF 2pF )
+           QNLB(TF)=OPT( 0.1NS 1NS 0.3NS )
+           QNLB(AF)=OPT( 0.2 5.0 1.0 )
+           QNLB(NF)=OPT( 0.1 3 1.0 )
+           QNLB(BR)=OPT( 0.2 3 2.38 )
+           QNLB(NR)=OPT( 0.5 3 1.0 )
+           QNLB(VJE)=OPT( 0.1 1.5 1.44 )
+           QNLB(MJE)=OPT( 0.1 1.5 0.33 )
+           QNLB(VJC)=OPT( 0.25 2 0.99 )
+           QNLB(MJC)=OPT( 0.1 1 0.33 )
+           QNLB(VJS)=OPT( 0.2 1.5 1.39 )
+           QNLB(MJS)=OPT( 0.1 2 0.5 )
+           QNLB(IS)=OPT( 1e-18 1e-14 92.51e-16 )
+           QNLB(KF)=OPT( 1e-18 1e-14 5.4e-16 )
+           QNLB(TR)=OPT( 2ns 15ns 6NS )
+           QNLB(CJE)=OPT( 0.5pF 10pF 3PF )

```

```
+          QNLB(CJC)=OPT( 0.5pF 10pF 2PF )
+          QNLB(VA)=OPT( 10 100 50 )

+          QNLC(BF)=OPT( 10 200 80 )
+          QNLC(RB)=OPT( 10 500 100 )
+          QNLC(CCS)=OPT( 0.5pF 10pF 2pF )
+          QNLC(TF)=OPT( 0.1NS 1NS 0.3NS )
+          QNLC(AF)=OPT( 0.2 5.0 1.0 )
+          QNLC(NF)=OPT( 0.1 3 1.0 )
+          QNLC(BR)=OPT( 0.2 3 2.38 )
+          QNLC(NR)=OPT( 0.5 3 1.0 )
+          QNLC(VJE)=OPT( 0.1 1.5 1.44 )
+          QNLC(MJE)=OPT( 0.1 1.5 0.33 )
+          QNLC(VJC)=OPT( 0.25 2 0.99 )
+          QNLC(MJC)=OPT( 0.1 1 0.33 )
+          QNLC(VJS)=OPT( 0.2 1.5 1.39 )
+          QNLC(MJS)=OPT( 0.1 2 0.5 )
+          QNLC(IS)=OPT( 1e-18 1e-14 92.51e-16 )
+          QNLC(KF)=OPT( 1e-18 1e-14 5.4e-16 )
+          QNLC(TR)=OPT( 2ns 15ns 6NS )
+          QNLC(CJE)=OPT( 0.5pF 10pF 3PF )
+          QNLC(CJC)=OPT( 0.5pF 10pF 2PF )
+          QNLC(VA)=OPT( 10 100 50 )

+          QNLD(BF)=OPT( 10 200 80 )
+          QNLD(RB)=OPT( 10 500 100 )
+          QNLD(CCS)=OPT( 0.5pF 10pF 2pF )
+          QNLD(TF)=OPT( 0.1NS 1NS 0.3NS )
+          QNLD(AF)=OPT( 0.2 5.0 1.0 )
+          QNLD(NF)=OPT( 0.1 3 1.0 )
+          QNLD(BR)=OPT( 0.2 3 2.38 )
+          QNLD(NR)=OPT( 0.5 3 1.0 )
+          QNLD(VJE)=OPT( 0.1 1.5 1.44 )
+          QNLD(MJE)=OPT( 0.1 1.5 0.33 )
+          QNLD(VJC)=OPT( 0.25 2 0.99 )
+          QNLD(MJC)=OPT( 0.1 1 0.33 )
+          QNLD(VJS)=OPT( 0.2 1.5 1.39 )
+          QNLD(MJS)=OPT( 0.1 2 0.5 )
+          QNLD(IS)=OPT( 1e-18 1e-14 92.51e-16 )
+          QNLD(KF)=OPT( 1e-18 1e-14 5.4e-16 )
+          QNLD(TR)=OPT( 2ns 15ns 6NS )
+          QNLD(CJE)=OPT( 0.5pF 10pF 3PF )
+          QNLD(CJC)=OPT( 0.5pF 10pF 2PF )
+          QNLD(VA)=OPT( 10 100 50 )

+          QNLE(BF)=OPT( 10 200 80 )
+          QNLE(RB)=OPT( 10 500 100 )
+          QNLE(CCS)=OPT( 0.5pF 10pF 2pF )
+          QNLE(TF)=OPT( 0.1NS 1NS 0.3NS )
+          QNLE(AF)=OPT( 0.2 5.0 1.0 )
+          QNLE(NF)=OPT( 0.1 3 1.0 )
+          QNLE(BR)=OPT( 0.2 3 2.38 )
```

```

+          QNLE (NR)=OPT ( 0.5  3  1.0 )
+          QNLE (VJE)=OPT ( 0.1  1.5  1.44 )
+          QNLE (MJE)=OPT ( 0.1  1.5  0.33 )
+          QNLE (VJC)=OPT ( 0.25  2  0.99 )
+          QNLE (MJC)=OPT ( 0.1  1  0.33 )
+          QNLE (VJS)=OPT ( 0.2  1.5  1.39 )
+          QNLE (MJS)=OPT ( 0.1  2  0.5 )
+          QNLE (IS)=OPT ( 1e-18  1e-14  92.51e-16 )
+          QNLE (KF)=OPT ( 1e-18  1e-14  5.4e-16 )
+          QNLE (TR)=OPT ( 2ns  15ns  6NS )
+          QNLE (CJE)=OPT ( 0.5pF  10pF  3PF )
+          QNLE (CJC)=OPT ( 0.5pF  10pF  2PF )
+          QNLE (VA)=OPT ( 10  100  50 )

+  TARGETS  gain_min=119 gain_max=121
+  OPTIONS  AVG=1e-6 NUMFUNC=500 NUMITER=300
+           MEASOFF=1 IPLOFF=1 MARQUP=5000
*
***** Select the optimizer
.OPTION OPTIMIZER=SA
.IPLOT VM(16)
.END

```

図4: 120個の設計変数を持つ広帯域増幅器のSmartSpiceネットリスト




図5: Simulated Annealing法による最適化前および最適化後の増幅器のゲインのプロット

デザイン #3 デジタル・インバータ

この例では、図 6 に示すデジタル・インバータにおいて、1.95V の出力電圧を持つという制約条件を満たしながら、供給電力を最小化する場合を考えます。

この例では、設計変数は電源電圧であり、ターゲット目標値は、低く設定された、DC 電源から供給される電力計算値と、1.95V の出力電圧です。ネットリストを図 7 に示します。2.5V の電源電圧 (V_SUPPLY) から開始すると、Levenberg-Marquardt オプティマイザは、ローカル・ミニマムに陥ってしまいます。

そのため、代わりにハイブリッド・オプティマイザを使用します。次のようにオプションを選択します。

```
.OPTION OPTIMIZER=H
```




図6: デジタル・インバータ

[環境設定] ダイアログの [オプティマイザ] 設定ページで、1 つ目のオプティマイザとして Parallel Tempering 法、2 つ目のオプティマイザとして Levenberg-Marquardt 法を選択すると、今回のターゲット目標値を満たす十分な解を見つけることができます。最適化前および最適化後の出力電圧を示す過渡結果を、図 8 に示します。

```
* Digital inverter

***** SOME PARAMETERS
.PARAM INP_FREQ      = 850MegHz
.PARAM INP_PERIOD    = 1/INP_FREQ
.PARAM NO_PERIODS    = 4
.PARAM TMEAS_START   = `(NO_PERIODS-1)*INP_PERIOD`
.PARAM TMEAS_STOP    = NO_PERIODS*INP_PERIOD
.PARAM T1             = `TMEAS_STOP - 1*INP_PERIOD/4`
.PARAM T2             = `TMEAS_STOP - 0.75*INP_PERIOD/4`

***** OPTIMIZATION PARAMETERS
.PARAM V_SUPPLY      = 2V
.PARAM WP            = 1000u
.PARAM LMIN          = 0.35u

***** SUPPLY VOLTAGES
VDS (VDD 0) DC V_SUPPLY

***** INPUT SIGNAL
VSIG (IN 0) PULSE(V_SUPPLY 0
+           INP_PERIOD/2
+           INP_PERIOD/1000 INP_PERIOD/1000
+           INP_PERIOD/2 INP_PERIOD)

***** OUTPUT STAGE
MP (OUT IN VDD VDD) pm W=WP L=LMIN
MN (OUT IN 0 0) nm W=WP/3 L=LMIN

***** Models
.MODEL pm PMOS ( level=3 tox=.02e-6
+ phi=0.576 gamma=0 vto=0 alpha=0 kappa=0 is=0)
.model nm NMOS ( level=3 tox=.02e-6
+ phi=0.576 gamma=0 vto=0 alpha=0 kappa=0 is=0)

CL (OUT 0) 10pF
```

continued...

```

**** Analysis statement
.TRAN INP_PERIOD/250 NO_PERIODS*INP_PERIOD CALLV SAVEV

.OPTIONS iplot_one nomod
.IPLOT v(OUT)

**** Measure statements
.MEASURE TRAN VHIGH MAX v(OUT) FROM=T1 TO=T2
.MEASURE TRAN PSUPPLY MAX '(v(VDD)*i(VDS))' FROM=T1 TO=T2

.MODIF
+ OPTIMIZE V_SUPPLY=opt(1 5 2.5)
+ TARGETS PSUPPLY=1e-6 VHIGH=1.95
* VLOW=0.001
+ OPTIONS AVG=1e-6 MEASOFF=1 NUMFUNC=50 NUMITER=50
+ IPLOFF=2

**** Select the optimizer
.OPTION OPTIMIZER=H

.END

```

図7: デジタル・インバータのSmartSpiceネットリスト




図8: 最適化前および最適化後のデジタル・インバータの出力電圧のプロット

まとめ

SmartSpice の最適化機能を使用することで、回路の設計を改良し、与えられた仕様を満たしたり改善したりできます。初期値と特定のターゲット目標値のセットに基づいて、最適化は、最適解が得られるまでパラメータ空間上で反復処理を行います。ローカル・最適化とグローバル・最適化の両方が利用可能です。ローカル・最適化は、開始点がローカル・ミニマムに近い場合に有効で、一般的

に高速です。グローバル・最適化は、より広い範囲のパラメータを探索できるため、グローバルな解に到達できる可能性があります。回路は、定常状態、周波数領域、時間領域など、利用可能な任意の解析手法を用いて最適化できます。SmartSpice にグローバル最適化機能が加わったことにより、多数のパラメータを用いてデザインを最適化できるようになり、よりフレキシブルな設計が可能になりました。